## 第5章

# 表計算ソフトウエア:少し複雑な処理 (lookup 関数,絶対参照, IF 関数)

2024年6月24日

## 学習目標

- (1) 絶対参照とスピルの考え方を理解する.
- (2) ソートの考え方を理解する.
- (3) IF 関数を使って,条件分けをできるようになる.
- (4) 表引き(LOOKUP 関数)の考え方を理解する.
- (5) さまざまな関数 (ROUNDUP,ABS など)の使い方を理解する.
- (6) 少し複雑な計算式になれる.
- (7) 絶対参照,相対参照が混じったものやそのスピルでの計算方法を理解する.

本章は、専修大学商学部の高萩栄一郎の著作である.

## 1 表計算ソフトウエアのしくみ

計算式を複写するとき,複写元の計算式中セルの番地は,複写に伴う番地の移動に伴い,複写先の計算式中の番地は自動 的に変化します.絶対参照(または絶対番地)とは,複写に伴いセルが移動しても変化させない(固定する)番地の指定方 法です.逆に,セルの移動とともに変化させるセルの番地の指定を相対参照(または相対番地)といいます.プログラミン グとの対応では,ループの中での定数が絶対参照,ループの中の配列の要素が相対参照に対応します.

表計算ソフトウエアは,複写すると相対参照でアドレスが調整されます.しかし,それでうまくいかないことがあります. そのときは、上記の絶対参照という機能とその範囲の計算式をまとめて設定するスピルという機能があります.

## 2 複写のしくみ

図1は、GDPの表を複写で設定した計算式で、次のことがわかります.

- S2:T12(図 1 の青の部分) 式がかかれているセルの数字が 1 増えると,計算式中の数字も 1 増えています.たとえば,S2 の計算式は「=SUM(B2:R2)」になっており,下に複写した S3 の計算式は「=SUM(B3:R3)」と,すべて数字が 1 増えています.
- B12:R13(図 1 の赤の部分 (一部)) 式が書かれているセルのアルファベットが 1 つ増えると,右に複写された計算式のアルファベットも 1 つ増えています.たとえば,P13 の計算式は「=AVERAGE(P2:P11)」になっており,右に複写した Q13 の計算式は,「=AVERAGE(Q2:Q11)」とアルファベットが 1 つ増えています.

したがって、「複写」という機能は、複写元のセルの番地と複写先のセルの番地との差の分、計算式中の番地を調整してい ることがわかります.別の言い方をすれば、計算式は設定しているセルからの相対的な位置で記録され、これは相対参照と

	А	Р	Q	R	S	Т	
1	Country	2015	2016	2017	合計	平均	衫
2	Australia	1232.913	1264.524	1379.548	=SUM(B2:R2)	=AVERAGE(B2:R2)	
3	Canada	1559.623	1535.768	1653.043	=SUM(B3:R3)	=AVERAGE(B3:R3)	
4	China	11226.186	11221.836	12014.61	=SUM(B4:R4)	=AVERAGE(B4:R4)	Ĵ
5	India	2102.392	2273.556	2602.309	=SUM(B5:R5)	=AVERAGE(B5:R5)	数
6	Japan	4394.977	4950.073	4873.202	=SUM(B6:R6)	=AVERAGE(B6:R6)	
7	Korea	1382.764	1414.804	1540.458	=SUM(B7:R7)	=AVERAGE(B7:R7)	カ
8	United Kingdom	2897.06	2669.107	2628.41	=SUM(B8:R8)	=AVERAGE(B8:R8)	増
9	United States	18219.3	18707.15	19485.4	=SUM(B9:R9)	=AVERAGE(B9:R9)	え
10	EURO area	11686.259	11960.698	12632.374	=SUM(B10:R10)	=AVERAGE(B10:R10)	7
11	Asean-5	2047.966	2148.331	2317.146	=SUM(B11:R11)	=AVERAGE(B11:R11)	ר <sup>י</sup>
12	合計	=P2+P3+P4+P5+P6+P	=Q2+Q3+Q4+Q5+Q6+0	=R2+R3+R4+R5+R6+F	=SUM(B12:R12)	=AVERAGE(B12:R12)	_ <
13	平均	=AVERAGE(P2:P11)	=AVERAGE(Q2:Q11)	=AVERAGE(R2:R11)			

複写するとアルファベットが増えていく

図1 相対参照のしくみ(計算式を表示)

呼ばれています. たとえば, 図1のS2の計算式は,「=SUM(B2:R2)」になっています. この計算式は,

= SUM ((自分より17つ左のセル)から(自分より1つ左のセル)まで)

というしくみになっています.多くの表は,この複写の番地調整機能(相対参照)でうまくいくのですが,うまくいかない ことがあります.その例をやってみましょう.

■参考:計算式の表示方法 計算式を設定するとその計算結果(数値など)が表示されます.計算結果に替えて,計算式を表示することもできます.

• もう一度 数式の表示 をクリックすることにより元に戻る

## 3 絶対参照,相対参照とスピル

## 3.1 相対参照だけではうまくいかない例

図2は,各支店の年度別の売上高を記入したものです.この表は,教科書のページからダウンロードできます (S6.xlsx). オレンジの背景色の部分は,計算式で入力する部分です

	А	В	С	D	Е	F	G
1		A支店	B支店	C支店	D支店	E支店	合計
2	売上高(2010)	1200	1800	2200	1100	1200	7500
3	売上高(2011)	1300	1800	2100	1200	1250	7650
4	伸び率	8%	0%	-5%	9%	4%	2%
5	構成比率(2011)	17%	24%	27%	16%	16%	100%
6	伸び率評価1	1	0	0	1	1	
7	伸び率評価2	1	0	0	1	1	

図2 売上高の評価

伸び率を計算しましょう.伸び率の計算式は,

2011年の売上高 - 2010年の売上高

2010年の売上高

で,表示形式は「パーセント」です.

次に、2011年の各支店の売り上げが、合計に占める割合(構成比率)を計算しましょう.

× B5: =B3/G3

となります. この計算式を B5 から F5 に, 複写して見ましょう. おかしな結果になります. 原因は表1の「単純に複写した ときの計算式」(相対参照)になってしまうからです. この場合, C5 から F5 の計算式がおかしいことがわかります. 分子は 複写先のセルの番地に従って, セルの番地は, 相対参照(番地調整機能)によってアルファベットが1つずつ増えてよいの ですが. 分母の番地はいつも「G3」のままでなくてはなりません(青).

これを解決するために, セル G3 を変化させない方法 (絶対参照) と B5 から F5 の計算式をまとめて設定することにより, G3 を変化さえない方法があります(スピル機能)

表1 セルの設定

セル	B5	C5	D5	E5	F5
単純に複写したときの計算式	=B3/ <mark>G3</mark>	=C3/ <mark>H3</mark>	=D3/ <mark>I3</mark>	=E3/ <mark>J3</mark>	=F3/ <mark>K3</mark>
正しい計算式	=B3/ <mark>G3</mark>	=C3/ <mark>G3</mark>	=D3/ <mark>G3</mark>	=E3/ <mark>G3</mark>	=F3/ <mark>G3</mark>

#### 3.2 絶対参照

表1の正しい計算式のように, 複写したときに G3 を変化させないようにするのが絶対参照です. 図 3 のように, 複写元 (B5)のセルを設定するとき, 番地調整機能を働かせないようにします. これは, 変化させないアルファベットや数字の前 に,「\$」マークをつけます. したがって, 複写元の計算式は, 次式になります.

## B5: =B3/\$G3

この場合, G3 は、常に G3 で、数値の方は変化しないので、

B5: =B3/\$G\$3

としてもかまいません. \$をつけた番地を「絶対参照」, \$を付けない番地を「相対参照」といいます(図3).



図3 絶対参照を使って設定した計算式

■絶対参照の指定の仕方 他の計算式と同様に,キーボードなどから\$マークを入力することが可能ですが,セルをクリック したあとに,簡単な操作で\$マークを入力する方法があります.

- セルの番地(G3など)をクリックなどで入力します.
- キーボードの(F4)キーを押すことにより、\$マークを付けることができます。
- \$ の位置が異なるときは、(F4)キーを何回か押すことにより、指定の計算式にします.

動画:絶対参照(音声付き)・動画:絶対参照(音声なし)

## 3.3 スピルの利用

Excel 2021 で追加された スピル という機能で、絶対参照を使って指定した範囲の計算式をまとめて指定することができ ます. まとめて式を設定する範囲を考えれば、絶対参照より直観的に理解できます.

スピルは、デスクトップ版の Excel 2019 などの古いバージョンの Excel では利用できません、その場合、スピルを使うと きは、VDI を使用してください.

スピルは、ある範囲のセルの計算をまとめて設定します.

■スピルの図解 図4は、伸び率の計算を例です.図4上段は、これまでのやり方でやったものです.

B4:	=(B3-B2)/B2
複写元:	B4
複写先:	C4:G4

通常の利用ではこの方法を利用しましたが、スピルのイメージを把握するため、下段のようにスピルを使って設定します. 計算に利用する範囲は2つ、計算結果(出力範囲)が1つになります.

範囲 P:	B2:G2	2010 年の値	
範囲 Q:	B3:G3	2011 年の値	
出力範囲 R:	B4:G4	伸び率	
スピルの計算式 (B4 に設定)	=(B3:G3-B2:G2)/B2:G2	B4 に設定すれば,	B4:G4 に出力される

| (2011 年の値 - 2010 年の値) / 2010 年の値 |ですので,| (Q-P)/P |となります.スピルでは,範囲の各値に対して順番 計算は.| に計算していきます (各範囲の1番目の値で計算した結果は、出力範囲の1番目に出力). 出力範囲は、入力範囲に応じて自

## 計算式の複写

	А	В	С	D	E	F	G
1		A支店	B支店	C支店	D支店	E支店	合計
2	売上高(2010)	1200	1800	2200	1100	1200	7500
3	売上高(2011)	1300	1800	2100	1200	1250	7650
4	伸び率	8%	0%	-5%	9%	4%	2%
5	構成比率(2011)	L					
6	伸び率評価1						
7	伸び率評価2	) =(	(B3-B2)/B	2			
0							

スピル





動的に拡張して表示されます(この拡張するのを、「こぼれ落ちる」ように増やすので、Spill と呼ばれています).

■スピルの計算式の設定(B4:G4) 計算式は、次のように設定します.

(1) | B4 |をクリックし、 (三)、(一)を入力します.

(2) マウスもしくはキーボードで B3:G3 (範囲 Q) を範囲指定し, - を入力

- (3) B2:G2(範囲 P)を範囲指定し,①⑦を入力
- (4) B2:G2(範囲 P)を範囲指定し,(Enter)を入力
- (5) 計算結果が B4:G4(範囲 R) に表示されます.

動画:スピルによる設定(伸び率)(音声付き)・動画:スピルによる設定(伸び率)(音声なし)

■構成比率の計算 図5は、スピルを使った構成比率の計算です.



図5 スピルによる構成比率の計算

範囲 S:B3:F32011 年の値セル T:G3合計の値,範囲 S の何番目かにかかわらず,いつもG3出力範囲 U:B5:F5構成比率スピルの計算式(B5 に設定)=B3:F3/G3B5 に設定すれば, B5:F5 に出力される

範囲 S(B3:F3, G3 は含めない)の値を 合計の値のセル (G3) で割って, S の各値の構成比率を求めます.構成比率は,

B3/G3, C3/G3,...のように、 範囲 S / セル T になります. 分母は, 範囲 S の何番目かにかかわらずいつも G3 です. スピ ルの場合, 1つのセルを指定した場合, いつもそのセルで計算します (これは, 絶対参照で\$を付けるのに対応します).

B5: =B3:F3/G3

(1) まとめて計算式を指定する範囲の左上のセルを選択します (B5)

(2) 次の計算式を入力して(B3:F3 は範囲指定を利用した方が便利), (Enter)

## B5: =B3:F3/G3

(3) B3:F5 は, B3 から右へ 5 個のセルなので, 計算式を入力した B5 から F5 までの 5 個のセルに こぼれる (spill) よう に, 計算結果が表示されます.

動画:スピルによる設定(構成比率)(音声付き)・動画:スピルによる設定(構成比率)(音声なし)

#### ■注意

- •「#SPILL」「#スピル」というエラーが表示されることがあります.これは,拡張された出力範囲にすでに値が入力さ れている場合に表示されます(その範囲を一度クリアするか,別の場所を指定します).
- スピルで設定した計算式を変更するときは、左上のセルの計算式を変更します.

## 4 指数を計算

GDP の計算で,経済発展が大きい国はどこであるのか分析するとき,国の大きさに関わらず,GDP がどのように変化しているのかを分析することが重要です.そこで,物価指数のように基準年を設け,各国について基準年の値を100としたと

き各年の GDP がいくつかになるのかを分析します.

1	А	В	С	D	E	F	G	н	1
1	GDP指数 2001年基準	2001	2002	2003	2004	2005	2006	2007	2008
2	Australia	100.00	112.65	143.24	174.14	194.84	207.22	251.38	279.93
3	Canada	100.00	102.93	121.19	138.94	158.80	178.63	198.94	210.36
4	China	100.00	109.93	124.33	146.29	171.78	206.41	265.72	342.56
5	India	100.00	106.04	125.19	146.09	168.89	192.15	250.78	247.83
6	Japan	100.00	95.62	103.30	111.89	110.50	105.27	104.92	117.06
7	Korea	100.00	114.25	127.66	143.49	168.49	189.81	210.61	188.02
8	United Kingdom	100.00	109.20	125.79	147.87	155.44	166.09	189.72	180.46
9	United States	100.00	103.35	108.28	115.42	123.20	130.55	136.57	139.04
10	EURO area	100.00	109.08	134.40	154.05	159.99	169.76	195.69	215.39
11	Asean-5	100.00	113.64	129.07	144.53	160.78	193.70	230.50	270.35
12	全体	100.00	104.77	117.60	130.84	138.86	147.80	164.10	175.99

#### 図6 指数化した表

図 6 は、その計算表で新しく表(シート名:GDP\_index)を作成しました.計算式は、2001 年のデータを 100 として、その何倍になったかを計算するので、

となります.

■絶対参照による計算 B2の計算式は、ワークシート間の演算を利用して「=(GDP!B2/GDP!B2)\*100」ですが、複写する とき、2000 年の GDP はいつも B 列なので、B を変化させないように B の前に\$をつけ、2 の前には付けません. (F4)キー を使うとき分母の B2 をクリックしたあと, (F4)キーを 3 回押すことにより!B2 になります. 2000 年から 2007 年までまとめて,計算式を複写します.



とします. B 列も同じ計算式で,100 となります. (F4)キーは,1回押す毎に\$マークが付く位置(アルファベットの前,数字の前)を変更しますので,何回か押して正しい位置になるようにしてください.動画:Excel GDP の指数

■スピルによる計算 図7は、スピルを使った各国地域の指数の計算です.

範囲 V を範囲 W で割り 100 倍する計算を指定しています. V の各値を W のどの値で割るのかはこの指定では自動で調整されます.具体的には,範囲 X の各セルで,分子の範囲 V のセルは,対応セルですが,分母の範囲 W では 1 列のみですので,範囲 X の同じ行の計算式は共通に範囲 W の同じ行の値を使っています.これは絶対参照と同じで,範囲をまとめて設定することにより\$を付ける位置を自動で調整していると考えることもできます.

- ワークシート GDP\_index を選択(または作成)
- まとめて指定する範囲の左上のセルを選択します (B2)
- B2 に (三)を入力
- GDPのワークシートで、B2:R12を範囲指定し、
- GDP のワークシートで, B2:B12 を範囲指定
- •「\*100」を入力し, (Enter)
- 計算式は次のようになります.

	ークシー	ト:G	DP		<b></b> 列E~	~Sは省略		
	А	В	С	D	Т	U	V	W
1	Country 🗸	2001	2002	2003	2019	2020	2021 슴	≣†
2	Australia	376.71	424.43	539.56	1,392.33	1,359.37	1,610.56	23,495.45
3	Canada	738.97	760.15	895.60	1,741.58	1,644.04	2,015.98	31,060.89
4	China	1,333.65	1,465.83	1,656.96	14,340.60	14,866.74	16,862.98	161,404.64
5	India	VV 493.93	523.77	618.37	2,870.50	2,660.24	2,946.06	35,451.26
6	Japan	4,374.71	4,182.85	4,519.56	5,135.90	5,045.10	5,103.11	105,453.66
7	Korea	547.74	626.99	702 70	1.651.42	1,638.26	1,823.85	25,780.19
8	United Kingdom	1,639.08	1,782.89	2,052.79	2,833.30	2,709.68	3,108.42	55,110.55
9	United States	10,581.93	10,929.10	11,456.45	21,372.60	20,893.75	22,939.58	339,171.98
10	Euro AREA	6,584.84	7,178.67	8,842.34	13,408.86	12,996.32	14,517.56	250,874.94
11	ASEAN 5	515.88	585.69	664.87	2,734.23	2,602.96	2,821.32	35,847.53
12	合計	27,187.44	28,460.35	31,949.19	67,481.32	66,416.46	73,749.41	1,063,657.08
13	半均	2,/18./4	2,846.04	3,194.92	6,748.13	6,641.65	7,374.94	
ワ	ークシー	ト:Gl	JP_ind	ex r	——————————————————————————————————————	E~Sは省	略	
	А	В	С	D	Т	U	V	
	GDP指数	2001	2002	200	2 201	10 20	20 20	21
1	2001年基準 〉	2003	2002	200	201	19 20	20 20	21
2	Australia	100.00	112.67	142.23				
3	Canada	100.00		145.2	369.60	0 360.8	35 427.5	3
5		100.00	102.87	143.2	3 369.60 235.68	0 360.8 8 222.4	35 427.5 18 272.8	
4	China	100.00	102.87 109.91	143.2 121.20 124.24	3 369.60 ) 235.68 I 1075.29	0 360.8 8 222.4 9 1114.7	35         427.5           48         272.8           74         1264.4	$\frac{3}{1}$ ( V
4	China India	100.00 100.00 100.00	102.87 109.91 106.04	143.23 121.20 124.24 125.19	3 369.60 235.68 4 1075.29 9 581.19	0 360.8 8 222.4 9 1114.7 5 538.5	35         427.5           48         272.8           74         1264.4           58         596.4	$\begin{array}{c}3\\1\\2\\5\end{array}$
4 5 6	China India Japan	100.00 100.00 100.00 100.00	102.87 109.91 106.04 95.61	143.23 121.20 124.24 125.19 103.33	3 369.60 235.68 1075.29 581.19 117.40	0 360.8 8 222.4 9 1114.7 5 538.5 0 115.3	35         427.5           48         272.8           74         1264.2           58         596.2           32         116.6	$\begin{array}{c}3\\1\\2\\5\\5\end{array}$
4 5 6 7	China India Japan Korea	100.00 100.00 100.00 100.00 100.00	102.87 109.91 106.04 95.61 114.47	143.2. 121.20 124.24 125.19 103.33 128.29	3         369.60           0         235.68           4         1075.29           581.11         117.40           0         301.50	0 360.8 8 222.4 9 1114.7 5 538.5 0 115.3 0 299.0	35         427.5           48         272.8           74         1264.4           58         596.4           32         116.6           09         332.5	$\begin{array}{c}3\\1\\2\\5\\5\\8\end{array}$
4 5 6 7 8	China India Japan Korea United Kingdom	100.00 100.00 100.00 100.00 100.00	102.87 109.91 106.04 95.61 114.47 108.77	143.2. 121.20 124.24 125.19 103.33 128.29 125.24	3         369.60           0         235.68           4         1075.29           9         581.19           1         117.40           9         301.50           4         172.80	0 360.8 8 222.4 9 1114.7 5 538.5 0 115.3 0 299.0 6 165.3	35         427.5           48         272.6           74         1264.2           58         596.2           32         116.6           09         332.5           32         189.6	$\begin{array}{c}3\\1\\2\\5\\5\\8\\4\end{array}$
4 5 6 7 8 9	China India Japan Korea United Kingdom United States	100.00 100.00 100.00 100.00 100.00 100.00	102.87 109.91 106.04 95.61 114.47 108.77 103.28	143.2. 121.2( 124.24 125.1( 103.3) 128.2( 125.24 108.2( 108.2(	3         369.60           0         235.68           4         1075.29           0         581.19           1         117.40           0         301.50           4         172.86           5         201.97	0 360.8 8 222.4 9 1114.7 5 538.5 0 115.3 0 299.0 6 165.3 7 197.4	35         427.5           48         272.6           74         1264.2           58         596.2           32         116.6           09         332.5           32         189.6           15         216.7	$\begin{array}{c}3\\1\\2\\5\\5\\8\\4\\8\end{array}$
4 5 6 7 8 9	China India Japan Korea United Kingdom United States Euro AREA	100.00 100.00 100.00 100.00 100.00 100.00 100.00	102.87 109.91 106.04 95.61 114.47 108.77 103.28 109.02	143.2 121.2 124.2 125.1 103.3 128.2 125.2 108.2 108.2 134.2 134.2 143.2 143.2 125.2 103.3 128.2 125.2 12	3         369.60           0         235.68           4         1075.29           0         581.19           1         117.40           0         301.50           4         172.88           5         201.97           3         203.63	0 360.8 8 222.4 9 1114.7 5 538.5 0 115.3 0 299.0 6 165.3 7 197.4 3 197.3	35         427.5           18         272.6           74         1264.4           58         596.2           32         116.6           39         332.5           32         189.6           15         216.7           37         220.4	$\begin{array}{c}3\\1\\2\\5\\5\\8\\4\\8\\7\end{array}$
4 5 6 7 8 9 10 11	China India Japan Korea United Kingdom United States Euro AREA ASEAN 5	100.00 100.00 100.00 100.00 100.00 100.00 100.00 100.00	102.87 109.91 106.04 95.61 114.47 108.77 103.28 109.02 113.53	143.2 121.2 124.2 125.1 103.3 128.2 125.2 108.2 134.2 128.8 128.8	3         369.6(           0         235.6(           4         1075.2(           9         581.1(           1         117.4(           9         301.5(           4         172.8(           5         201.9(           3         203.6(           8         530.0()	0 360.8 8 222.4 9 1114.7 5 538.9 0 115.3 0 299.0 6 165.3 7 197.4 3 197.3 1 504.9	35         427.5           18         272.6           74         1264.4           58         596.2           32         116.6           09         332.5           32         189.6           15         216.7           37         220.2           56         546.6	$\begin{array}{c}3\\1\\2\\5\\5\\8\\4\\8\\7\\9\end{array}$

図7 スピルを使った各国地域の指数の計算

- 計算式で「GDP!B2:R12」とあるので、B列からR列まで17列、2行目から12行目まで11行、式を入力するセル
   B2から右下へ17列12行まとめて計算します。
- 分母の「GDP!B2:B12」は行のみの変化ですので、行のみの範囲を指定します.
- B2:B12 が, 2001 年基準(B列)を表します.

動画:スピル・指数の計算(音声付き)・動画:スピル・指数の計算(音声なし)

#### 4.1 練習問題 S1-6

スピルまたは絶対参照を利用して作成しましょう.

- (1) ワークシート「GDP\_index2008」を作成し、2008 年基準(リーマンショック)の GDP の指数の表を作成しなさい.
- (2) ワークシート「population\_index2001」を作成し、2001 年基準の人口の指数の表を作成しなさい.
- (3) ワークシート「GDPperCapita\_index2010」を作成し、2010 年基準の一人あたり GDP の指数の表を作成しなさい.

## 5 練習問題

## 5.1 練習問題 S7-1

図 8 は,関東の各都県の小学生,中学生,高校生,大学生の人数を千人単位で記入したものです.(出典:学校基本調査, 1999 年 5 月 1 日現在)この表は,教科書のページからダウンロードできます (S7.xlsx).

(1) B10 から E10 に、小学生、中学生、高校生、大学生の人数の合計を計算しなさい.

(2) F3 に、茨城県の小学生人数の関東の都県の合計に占める割合(パーセント表示)を計算します

	Α	B C		D	E	F	G	Н	I
1		人数(単位	:1000人)			各都県がと	ある割合		
2	都県名	小学生	中学生	高校生	大学生	小学生	中学生	高校生	大学生
3	茨城	189	110	106	37	8.55%	8.94%	8.58%	3.20%
4	栃木	128	74	77	21	5.79%	6.01%	6.23%	1.82%
5	群馬	124	69	67	19	5.61%	5.61%	5.42%	1.65%
6	埼玉	408	217	206	114	18.45%	17.63%	16.67%	9.87%
7	千葉	339	190	183	114	15.33%	15.43%	14.81%	9.87%
8	東京	562	329	365	660	25.42%	26.73%	29.53%	57.14%
9	神奈川	461	242	232	190	20.85%	19.66%	18.77%	16.45%
10	合計	2211	1231	1236	1155	100.00%	100.00%	100.00%	100.00%

図8 関東の県の児童生徒学生の比率

絶対参照またはスピルを使って指定しなさい.

絶対参照を使う場合 F3 の計算式を F3 から F9 に複写するために, F3 の計算式の適当な位置に\$マークをつけます.

ただし、\$マークは変化させないものの前につけます.

アルファベットを変化させたくない場合は,アルファベットの前,数字を変化させたくない場合は数字の前に\$を つけます.

F3 の計算式を F3 から F9 に複写します.

スピルを使う場合 各都道府県の小中高大学生の合計を sum 関数を使い、計算式の中に含めます.

スピルで, sum(範囲)を使うと, sum(範囲)は1つの値を出力する関数ですので,1つの値として扱われます. (3) 中学生,高校生,大学生の列も同様に計算式を設定します.

	A	В	С	D	E
1	合否判定				
2	合格基準点	70			
3					
4					
5	番号	名前	得点	平均との差	合否
6	1	田中	100	34.8	1
7	2	大西	70	4.8	1
8	3	大久保	30	-35.2	0
9	4	前田	80	14.8	1
10	5	友田	65	-0.2	0
11	6	保田	77	11.8	1
12	7	大川	78	12.8	1
13	8	木村	32	-33.2	0
14	9	村田	54	-11.2	0
15	10	川上	66	0.8	0
16		合計	652		5
17		平均	65.2		

図9 合否判定

5.2 練習問題 S8-1

図 9 のように入試の合否判定をします. この表は,教科書のページからダウンロードできます (S8.xlsx). ただし, E 列の 合否判定は, IF 関数の節で学習します.

- (1) 得点の合計, 平均を計算しましょう.
- (2) 平均との差を計算しましょう. 平均値はいつも C17 であることを注意しましょう.絶対参照またはスピルを使います.

## 5.3 練習問題 S1-7

各国・地域の GDP が各年,10 ヶ国・地域の合計に対しての割合(構成割合)を求めましょう.

- (1) 新しいシート「GDP\_share」を作成し,指数と同様に,表頭,表側を複写します.
- (2) A 国・地域のt 年の構成割合は,次式で計算します.

A国・地域 t 年の構成割合 =  $\frac{A 国 \cdot 地域 t 年 0 \text{ GDP}}{t \mp 0 10 \text{ } \text{r} \text{ I} \cdot \text{i} \text{ u} \text{ i} \text{ O} \text{ GDP} \text{ of } \text{chi}}$ 

B2~V11 に、A 国・地域 t 年の構成割合を計算します。絶対参照またはスピルを使います。

絶対参照 B2 に計算式を設定します. B2 の場合,分子は 2001 年の Australia の GDP(GDP!B2) になります.分母は,2001 年の GDP の合計 (GDP!B12) になりますが,複写すとき,合計はいつも 12 行目であることに注意しましょう.

#### 複写元: B2 複写先: B2:R11

スピル 分子は,指数の計算では 2001 年の値の列でしたが,構成割合では合計の行になることに注意しましょう. (3)表示形式をパーセンテージにします.

(4) 図 10 は,計算例です.

	A	В	С	D	E	F	G	Н
1	GDP構成割合	2001	2002	2003	2004	2005	2006	2007
2	Australia	1.39%	1.49%	1.69%	1.85%	1.95%	1.95%	2.13%
3	Canada	2.72%	2.67%	2.80%	2.89%	3.11%	3.29%	3.30%
4	China	4.96%	5.21%	5.25%	5.55%	6.14%	6.93%	8.04%
5	India	1.82%	1.85%	1.94%	2.04%	2.22%	2.37%	2.79%
6	Japan	15.89%	14.50%	13.96%	13.59%	12.64%	11.32%	10.16%
7	Korea	1.97%	2.15%	2.14%	2.16%	2.39%	2.53%	2.53%
8	United Kingdom	6.00%	6.26%	6.42%	6.79%	6.72%	6.75%	6.94%
9	United States	39.07%	38.54%	35.97%	34.46%	34.66%	34.51%	32.51%
10	EURO area	24.32%	25.32%	27.79%	28.63%	28.02%	27.93%	29.00%
11	Asean-5	1.86%	2.02%	2.04%	2.05%	2.15%	2.43%	2.61%

図 10 GDP の構成割合 (一部)

## 6 ソート (並べ替え) 関数

ソート(ソーティング,並べ替え)は,ある基準で行(または列)の順番を入れ換えることです.成績処理をするとき,得 点の大きい順に並べ替えるときなどに使います.GDPの例ではある年のGDPの大きさなどの順番で並べ替えてみます.例 題の表は,教科書のページからダウンロードできます (sort\_ex.xlsx).

2023 年度から本テキストでは Sort 関数を利用します. 従来のテキストでは,関数を使わずに,直接,表を書き換える方法でした. 直接書き換えると,分析の途中で元の表が書き換わり操作が複雑になったり,操作ミスにより元にデータが壊れたりしますので,ソート関数を使い. 別の表に出力するようにしました. 古いバージョンの Office では, Sort 関数は利用できません. その場合は VDI を利用してください.

#### 6.1 ソート関数の利用

Excel ではソートをする関数は,SORT 関数と SORTBY 関数の2つあります.SORT 関数は1基準のみで,SORTBY 関数は複数基準を扱えます.本テキストでは,複数基準を扱える SORTBY 関数を学修します.

ソート (並べ替え) は,ある基準に従って行を並べ替えます.並べ替えの方法は,大きい順は降順,小さい順 (または ABC 順,あいうえお順) は昇順と呼びます.並べ替えの基準の列ををソートキーもしくは単にキーといいます.

この節の例題は,教科書のページからダウンロードできます (sort\_ex.xlsx). 図 11 の元データ (出席簿のデータ)で,クラ ス,出席番号はばらばらの順番です.まず,真ん中の1基準のように,クラスの順番 (昇順)で並べ替えてみます.この場 合,クラスの順番は整いますが,出席番号はばらばらです.そこで,出席番号の大きい順 (降順)に並べます (通常は,小さ い順ですが,ここでは練習のため大きい順にします).このように,基準となる列が複数指定するとき,基本はクラス順で, クラスが同じ時とは出席番号順に並べ替えるとき,クラス順を1番目のソートキーにし,出席番号を2番目のソートキーに します.

SORTBY 関数は、元のデータの範囲を指定し、別の範囲にソートした結果を出力します。例えば、図 11 のワークシート 元データ の範囲 A2:A9 を別の範囲、図 11 のワークシート 1 基準 の A2:C19 に、ソート結果を表示します。関数は、左上 のセル A2 に指定すれば、C19 まで自動で拡張 (スピル) して表示されます。1 基準 の A2:C19 のように、スピルの範囲は スピル範囲と呼ばれています。

SORTBY 関数は次のように指定します.

= SORTBY(もとの範囲, 基準1の列の範囲, 基準1の順序, 基準2の列の範囲, 基準2の順序, ...)

元の範囲 並べ替える元の範囲 (図 11 の例では, 元データの A2:C19). 基準1の列の範囲 並べ替える基準の列の範囲. 行数は並べ替える元の範囲と同じ行数 (図 11 の 1 基準 の例では,

	А	В	С	D		А	В	С	D		А	В	С	D
1	クラス	出席番号	氏名		1	クラス	出席番号	氏名		1	クラス	出席番号	氏名	
2	В	1	生田花子		2	А	3	神保町四郎		2	А	3	神保町四郎	
3	С	1	秋葉原太郎		3	A	1	あざみ野菊代		3	A	2	飯田橋梅香	
4	А	3	神保町四郎		4	А	2	飯田橋梅香		4	А	1	あざみ野菊代	
5	А	1	あざみ野菊代		5	В	1	生田花子		5	В	3	登戸三郎	
6	В	2	向ヶ丘次郎		6	В	2	向ヶ丘次郎		6	В	2	向ヶ丘次郎	
7	С	2	水道橋桜美		7	В	3	登戸三郎		7	В	1	生田花子	
8	А	2	飯田橋 <mark>梅</mark> 香		8	С	1	秋葉原太郎		8	С	2	水道橋桜美	
9	В	3	登戸三郎		9	С	2	水道橋桜美		9	С	1	秋葉原太郎	
10					10					10				
<	>	元データ	1基準 … +		<	>	1基準 2基	[進 十	: ••	1	> <b></b>	1 其進 2 5	ŧ淮 ⊥	: 0

元のデータ

1基準(クラス昇順)

2基準(クラス昇順,出席番号降順)

図11 ソート関数の図解

基準1の順序 昇順の場合1,降順の場合-1を指定

図 11 のワークシート 1 基準 の A2 の計算式を設定します.

(2) = 「=SORTBY(」と入力

(3) = 元データのA2:C19を範囲指定し,「,」を入力

5-20

- (4) = 元データ の A2:A19 を範囲指定し,「,」を入力
- (5) = アルファベット順は昇順なので、「1」を入力
- (6) = 「)」を入力し, (Enter)を入力

#### =SORTBY(元データ!A2:C9, 元データ!A2:A9,1)

となります.

動画:Sortby 関数, 1基準(音声付き)・動画:Sortby 関数, 1基準(音声なし)

ワークシート2基準のA2の計算式は,次のようになります.基準1と同じ方法で設定してみましょう.手順の(3),(4) をくり返すのと,基準2の順序が降順であることに注意してください.

=SORTBY(元データ!A2:C9, 元データ!A2:A9,1, 元データ!B2:B9,-1)

ソート関数の特徴

- ソート関数は、元の表を書き直すと自動的にソート関数で出力されるスピル範囲は変化します。例えば、元データのクラス記号や出席番号を変更してみましょう。
   1 基準や2 基準の表が変化します。
- スピル範囲の個別のセルは変更できません.
   計算式を変更するには、左上のセル (A2)の計算式を変更します.
   スピル範囲を消すには、スピル範囲全体 (A2:C9)の中をクリックすると青色の枠で選択されるので、そこで(Delete) キーで削除します.

動画:Sortby 関数, 2基準(音声付き)・動画:Sortby 関数, 2基準(音声なし)

#### 6.2 練習問題 S1-8

ワークシート「GDP\_Sort」を作成し、2001 年を基準として大きい順にソートした結果を表示する表を作成しなさい. ※ソートする範囲は、列はデータの一番右まで含め、行は見出し行や合計や平均の行は含めません(含めると一緒にソート します).

## 7 RANK 関数を利用して順位を求める

S1.xlsx のワークシート GDP の AC 列のように, 2001 年の GDP での順位を求めましょう (図 12). 順位は, RANK 関数 という関数とスピルという Excel の新しい機能で求めることができます.

大学の PC の端末室に設置されている PC などで利用するときは,ブラウザー版の Excel を利用しましょう.

※ 表の画面をを書き直す必要あり

## 7.1 RANK 関数

図 13 は, RANK 関数の図解です. RANK 関数には, 3 つの引数が有ります.

数値 順位を求める値です.

範囲 この範囲で指定した範囲の順位を求めます.

順序 大きい順(降順)のとき 0, それ以外(小さい順 (昇順))のとき 0 以外の値を指定します.

	A	Т	U	V	W	Х	Y	Z	AA	AB	AC	AD	AE
1	Country	2019	2020	2021	合計	平均	中央値	GDP 2000→2021	平均倍数 幾何平均	平均増減率 (1年あたり増減率)	2001順位	2021順位	順位 平均増減率
2	Australia	1,392.33	1,359.37	1,610.56	23,495.45	1,118.83	1,251.86	4.2753	1.0753	7.53%	10	10	
3	Canada	1,741.58	1,644.04	2,015.98	31,060.89	1,479.09	1,556.51	2.7281	1.0515	5.15%	6	8	
4	China	14,340.60	14,866.74	16,862.98	161,404.64	7,685.94	7,492.21	12.6442	1.1353	13.53%	5	2	
5	India	2,870.50	2,660.24	2,946.06	35,451.26	1,688.16	1,823.05	5.9645	1.0934	9.34%	9	6	
6	Japan	5,135.90	5,045.10	5,103.11	105,453.66	5,021.60	5,003.68	1.1665	1.0077	0.77%	3	4	
7	Korea	1,651.42	1,638.26	1,823.85	25,780.19	1,227.63	1,253.42	3.3298	1.0620	6.20%	7	9	
8	United Kingdom	2,833.30	2,709.68	3,108.42	55,116.55	2,624.60	2,704.50	1.8964	1.0325	3.25%	4	5	
9	United States	21,372.60	20,893.75	22,939.58	339,171.98	16,151.05	15,599.73	2.1678	1.0394	3.94%	1	1	
10	Euro AREA	13,408.86	12,996.32	14,517.56	250,874.94	11,946.43	12,666.69	2.2047	1.0403	4.03%	2	3	
11	ASEAN 5	2,734.23	2,602.96	2,821.32	35,847.53	1,707.03	1,971.23	5.4689	1.0887	8.87%	8	7	
12	合計	67,481.32	66,416.46	73,749.41	1,063,657.08		53,972.31	2.7126	1.0512	5.12%			
13	平均	6,748.13	6,641.65	7,374.94									

図 12 RANK 関数(ワークシート GDP)



図 13 RANK 関数の図解

5-23

## 7.2 スピルを使って設定

計算式は,次のようになります.

## AC2: =RANK(B2:B11,B2:B12,0)

- RANK 関数の最初の引数は、1つの値なので、最初の引数の B2:B11 は、下に 10 個の分 (Y2:Y11) に計算することを 指定します.
- RANK 関数の2つ目の引数は,順位を求める範囲なので B2:B11 とします. (特に,「\$」を付ける必要はありません)

図 13 は RANK 関数の図解で RANK 関数は 3 つの引数を使います. AC2 にオーストラリアの 2001 年の順位を求めます. 次のように設定します.

- (1) AC2 をクリック
- (2) 「=RANK(」と入力
- (3) B2からB11を範囲指定し「,」を入力
- (4) B2からB11を範囲指定
- (6) 「)」を入力し, (Enter)を入力
- (7) AC2 から AC11 に 2001 年の順位が表示されたと思います.

動画:RANK 関数スピル(音声付き)・動画:RANK 関数スピル(音声なし)

## 7.3 絶対参照を使って設定

次のように設定します.

- (1) AC2 をクリック
- (2) 「=RANK(」と入力
- (3) B2 をクリックし、「,」を入力
- (4) B2からB11をマウスで範囲指定
- (5) キーボードからF4キーを1回入力.「B2:B11」が「\$B\$2:\$B\$11」となります.
- (6) 「)」を入力し, (Enter)を入力

複写



動画:Excel 順位を求める, 関数の挿入を利用

8 練習問題

## 8.1 練習問題 S1-9

(1) AD 列に 2021 の順位を求めましょう.

(2) ワークシート GDPperCapita の AC 列に 2001 年の, AD 列に 2021 年の一人あたりの GDP の順位を求めましょう.

#### 8.2 練習問題 S4-2

練習問題 S4-1 で作成したモーグルの表を「トータルスコア」の高い順に順位を付けなさい.

## 9 オートフィルター

オートフィルタは、ある特定の条件を満たす行だけ抽出する機能です.

### 9.1 オートフィルターの設定・解除

オートフィルターを実行すると、大きい順や小さい順に並べ替える機能がありますが、表を書き換えてしまうので利用し ないようにしてください.(ソートした結果は、フィルーターを解除しても戻らないようですので、設定した場合(Ctrl)+(こ)で 元に戻してください.

(1) 項目名の列を範囲してします. GDP の例では, (例: A1:X1) を範囲指定します.

- (2) メニューの データ  $\rightarrow$  並べ替えとフィルター の中の フィルター
- (3) A1:X1 の右に, |▼ |のボタンが表示されます.

■フィルターを使って,行の指定 次に,フィルターで扱う行を制限します(合計や平均の行を扱わないようにします) (1)項目名の右の▼のボタンをクリックします(ここでは,A1の右の▼ボタンをクリック) (2) メニューが表示されるので、扱わない行(ここでは、「合計」、「平均」、「空白のセル」のチェックを外します.
(3) OK をクリック

■抽出 オートフィルターを使うことにより,必要な行だけ抽出することができます.たとえば,2017年の GDP が 10000 百万ドルより多い行だけ選択してみます.

- (1) 基準の列(2017 の列)の ▼ のボタンをクリックします
- (2) 数値フィルターをクリックします,
- (3)「指定の値より大きい」を選び, 10000 を入力し, | 完了 |をクリック

**■フィルターの解除**使い終わったら、フィルターの解除を行ってください.

※フィルターの機能に,並べかえの機能がありますが,これを利用すると表を書き換えてしまうので,この演習では利用しないようしましょう.

## 10 IF 関数

## 10.1 IF 関数とは

S6 の売上高評価の表で,「伸び率評価1」の計算式を設定しましょう.これは,伸び率が正だったら1,0または負だった ら,0を表示させます.

あるときは、こうして、そうでないときは、こうするといったことをしたいときは、IF 関数を使います. 使い方は、

	А	В	С	D	Е	F	G
1		A支店	B支店	C支店	D支店	E支店	合計
2	売上高(2010)	1200	1800	2200	1100	1200	7500
3	売上高(2011)	1300	1800	2100	1200	1250	7650
4	伸び率	8%	0%	-5%	9%	4%	2%
5	構成比率(2011)	17%	24%	27%	16%	16%	100%
6	伸び率評価1	1	0	0	1	1	
7	伸び率評価2	1	0	0	1	1	

図 14 売上高の評価 (再掲)

IF(条件,条件を満たしているときすること,条件を満たしていないときすること)

となります. したがって,

## **B6:** =IF(B4>0,1,0)

とします. その意味は, 図 15 のようになります.

大小関係などの条件の表記は,数学の表記とは若干異なり,表2のようになります.

練習として,伸び率評価2の計算式を設定しましょう.伸び率評価2は,その支店の伸び率が全体の伸び率(セル G4 の 値)より大きいまたは等しければ1を,そうでなければ0を表示させます.絶対参照またはスピルを使いますので注意して ください.

### 10.2 講演会参加費

図 16 は、ある講演会の会費の計算をしたものです. この表は、教科書のページからダウンロードできます (S9.xlsx).



図 15 IF 関数の図解

表2 Excel での大小関係の表記

大小関係	Excel の表記
=	=
$\neq$	<>
>	>
$\geq$	>=
<	<
<	<=

会員/非会員 (B 列) 会員と非会員では、参加費が異なります.会員には、「M」と入力されています. 学生/非学生 (C 列) 学生には、参加費が割引(学生割引)されます.学生には、「S」と入力されています. パーティ (D 列) 講演会終了後に開かれるパーティに参加する人は、「P」と入力されています.

	Α	В	С	D	E	F	G	Н
1	NO	会員/非会員	学生/非学生	パーティ	参加費	学生割引	パーティ代	合計
2	1	М	S	Р	6,000	3,000	4,000	7,000
3	2		S		9,000	3,000	0	6,000
4	3			Р	9,000	0	4,000	13,000
5	4	М		Р	6,000	0	4,000	10,000
6	5	М	S		6,000	3,000	0	3,000
7	6	М	S		6,000	3,000	0	3,000
8	7	М	S		6,000	3,000	0	3,000
9	8	М		Р	6,000	0	4,000	10,000
10	9				9,000	0	0	9,000
11	10	М			6,000	0	0	6,000
12	11	М		Р	6,000	0	4,000	10,000
13	12	М			6,000	0	0	6,000
14	13	М			6,000	0	0	6,000
15	14	М	S	Р	6,000	3,000	4,000	7,000
16	15	М	S		6,000	3,000	0	3,000
17	16	М		Р	6,000	0	4,000	10,000
18	17			Р	9,000	0	4,000	13,000
19	18				9,000	0	0	9,000
20	19	М		Р	6,000	0	4,000	10,000
21	20	М		Р	6,000	0	4,000	10,000
22				合計	135,000	21,000	40,000	154,000
23		矣加弗(今吕∖	6.000					
24		<u> </u>	6,000					
26		<u> ジ加良(</u> 非云貝) 労止割引	9,000					
27		<u>ナエ剖り</u> パーティ	3,000					

#### 図 16 講演会会費

- 参加費 (E 列) 参加費を記入します. 参加費は変更されることもあるので, C24 もしくは C25 を参照するようにします.
- 学生割引 (F列) 学生割引を記入します. 学生割引は変更されることもあるので, C26 を参照するようにします.

5-30

パーティ代 (G 列) パーティ代を記入します. パーティ代は変更されることもあるので, C27 を参照するようにします. 合計 参加費 – 学生割引 + パーティ代 で計算します.

■参加費の計算(スピル)

E2: =IF(B2:B21="M",C24,C25)

動画:スピル・参加費の計算(音声付き)・動画:スピル・参加費の計算(音声なし)

■参加費の計算 (絶対参照) E2 のセルの計算式を設定します.参加費は, B2 が,「M」か空白によって異なるので IF 関数 を使います.

 $\triangle$  E2: = IF(B2="M",6000,9000)

B2 が「M」かどうか調べるのに,「B2="M"」とします. 文字は, ""で囲みます. M のとき, 6000 なので, 真の時の値を 6000 とし, 偽のときの値を 9000 としました. しかし, その 6000 と 9000 という値は, 変更されることがあり, C24 と C25 を参照するようにします. C24,C25 は, 下に複写されたときも固定される番地なので, 絶対参照にします. したがって,

E2: =IF(B2="M",\$C\$24,\$C\$25)

とし,下方向に複写します.

10.3 練習問題 S9-1

S9(参加費の計算)を完成させましょう.

#### 10.4 練習問題 S8-2

S8 の合否を設定します. 合格表示という列に, 基準点以上であれば1を, そうでなければ0を表示するようにします. もちろん, 合否の列は1箇所だけ計算式を作成して, あとは複写で計算式を設定してください. 合格表示の合計を算出すれば合格者数を求めることができます.

基準点を変えて,何人合格になるのか試してみようと思います.つまり,基準点を変えてみて,合格者が何人になるのか シミュレーションを行います.

## 11 表引き (LOOKUP 関数)

コンピュータを利用しているとよくコードを使います.例えば,大学生なら学生コード,商品には商品コードが付けられています.また,ある範囲での対応表から表を引くという作業があります.例えば,所得額から,税額を求めるといったことです. これらの機能を実現するのが VLOOKUP(または,HLOOKUP)という関数です.

JR 南武線 (路線図) の運賃計算の例で説明しましょう. 図 17 は,南武線の2つの駅名を入力して,川崎からの距離を求め, その差から,2駅間の距離を求め,それから運賃を求めるというものです.この表は,テキストのページからダウンロード できます (S101.xlsx(運賃計算),南武線は,電車特定区間の運賃).

図 17 の表は 3 つの表からできています.一番上が運賃を計算する表,二番目が駅名と川崎からの距離の対応表,三番目は,距離と運賃の対応表です.手順は次のようになります.

- (1) 乗車駅名から川崎からの距離を求めます (C2).
- (2) 降車駅名から川崎からの距離を求めます (C3).
- (3) 2 つの「川崎からの距離」の差をもとめ、小数点未満を切り上げます (C4,C5).

	A	В	С		
1		駅名	川崎からの距離		通信を
2	乗車駅	稲田堤			定長で
3	降車駅	登戸		<ul> <li>C</li> </ul>	
4	2駅間距離 切り上げ前				訂具
5	2駅間距離 切り上げ後				B 1 2 1
6	運賃(ICカードを利用)				
7	運賃(切符を利用)			)	
8					
9	駅名	川崎からの距離			駅名と川
10	川崎	0			
11	武蔵小杉	7.5			はないの
12	武蔵溝ノロ	12.7			呵からの
13	登戸	17.3		$\succ$	
14	稲田堤	20.8			品質の対
15	稲城長沼	24.1			エニービューン
16	府中本町	27.9			
17	分倍河原	28.8			いる
18	立川	35.5			
19					
20	距離	運賃(IC)	運賃(切符)		
21	1	1 3 3	140		明白肉化し
22	4	154	160		此内田へ
23	7	165	170		
24	11	216	220	<b>&gt;</b>	運賃の
25	16	302	310	(	定良い
26	21	388	390		<u>++</u> +++=
27	26	464	470		灯心衣
28	31	550	550		
29	36	637	640		
30	41			J	
31				-	

図 17 JR 南武線の運賃計算

(4) 距離と運賃の対応表から運賃をもとめます (C6,C7).

## 11.1 乗車距離から運賃を求める

まず乗車距離から運賃 (C6) を求めてみましょう. 仮に乗車距離として C5 に 4 を入力します. C6 を求めるには, 距離と 運賃の対応表を C5 の 4 について引くことです. 乗車距離は 4 なので表のロの行「4 154」の行があてはまり運賃は 154 円になります.

JRの運賃表				Exce	の表		
	距離	運賃(IC	)	距離	運賃(IC)		
イ	1 3	133		1	133	$\sum$	←1以上4未満のとき133円
	4 6	154		4	154		(距離は整数なので1~3のとき130円)
$\overline{\Lambda}$	7~10	165		7	165		
=	11~15	216		11	216		←11以上16未満のとき216円
ホ	16~20	302		16	302		(距離は整数なので11~15のとき216円)
^	21~25	388		21	388		
۲	26~30	464		26	464		
チ	31~35	550		31	550		
IJ	36~40	637		36	637		←36以上41未満のとき637円
				41			(距離は整数なので36~40のとき637円)

図 18 JR 運賃表と Excel の VLOOKUP 関数用の表との対応関係

11.1.1 対応表の見方

図 18 は, JR 運賃表を Excel の VLOOKUP 用の表には, どのように記述したらよいのかを示しています. Excel の表の各 行がどの数値をあてはまるかは, その行の距離の値から, 次の行の距離の値までです. 正確に言うと対応表の各行にあては まる数値は, その行の距離から次の行の値未満までです.

距離が1の行(イ) 距離が1以上4未満の運賃が133円であることを示しています. 距離が4の行(ロ) 距離が4以上7未満の運賃が154円であることを示しています

例えば,乗車距離が22であったら,「21 388」の行(へ)にあてはまります.

## 11.2 VLOOKUP 関数

C6 の計算式は,図 19 は,VLOOKUP 関数の図解です.

C6: =VLOOKUP(C5,A21:C30,2,TRUE)

VLOOKUP 関数は,VLOOKUP(照合値,照合範囲,列位置,照合方法)の4つの引数を書きます.(C5 に何も書いていないと エラーになるので,C5 に仮の2 駅間距離 (例えば6) を入力し,この値をいろいろ変えて見てください.)

照合値 表引きをする表で照合する値. この例では, 乗車距離 (C5)

(距離と運賃の対応表の一番左の列から探す値)

照合範囲 (対応表の範囲) 照合する表,ただし,照合値の列は,照合する範囲の一番左の列になるようにします. 列位置 照合した行の左から何番目の列を表示するか.一番左の列を1列目と数えます.

照合方法 完全に一致するもののみを探すのか (FALSE),下の行と未満のものも探すか (TRUE)を指定します.



図 19 VLOOKUP 関数の書き方

照合方法が TRUE のとき,表引きをする表は,一番左の列をソートキーに昇順 (小さい順) にソートされてなくてはなりません.理由は,照合するとき,照合値が,一番左の列の値と次の行の一番左の列の値の間にあるかどうかで,照合するかしないか決めているからです.

## C6: =VLOOKUP(B5,A21:C30,2,TRUE)

練習として, C7 の切符を利用した場合の運賃を計算しましょう.この場合 3 列目になることに注意しましょう. C5 の仮の値をいろいろ変更してみて,それぞれの運賃が正しく表示されているか確認しましょう.

動画:スピル・Vlookup 関数(音声付き) ・動画:スピル・Vlookup 関数(音声なし)

#### 11.3 駅名から距離を求める

次に,駅名から「川崎からの距離」を求める関数を求めましょう.駅名から距離を求めるときは,完全に一致する行のみを 探します.したがって,ソートしておく必要はありません. C2の計算式は,B2を照合値として,A10:B18の照合範囲を探 し,1列目と等しければ,2列目の値を返すというものです.完全に一致する行のみ探すので,「FALSE」にします.C3も 同様の計算式のため,絶対参照もしくはスピルを利用します.

C2 と C3 は,基本的に同じ計算式で,照合値が B2 か B3 かの違いで,照合範囲は同じで,列位置,照合方法は一定の値です.

#### ■スピルを利用するとき

### C2: =VLOOKUP(B2:B3,A10:B18,2,FALSE)

範囲 B2:B3(縦1列,2個)の値を照合値をスピルの範囲として使い,照合範囲 A10:B18 は共通で使います.出力は,照合範囲が縦1列,2個(2行)ですので,出力のスピル範囲も C2:C3 の縦1列,2個(2行)になります. 動画:スピル・駅名から距離(音声付き)・動画:スピル・駅名から距離(音声なし)

■絶対参照を利用するとき 絶対参照を利用するときは, 照合範囲が常に A10:B18 になるので, \$A\$10:\$B\$18 のように, 絶対参照にし, C2 を C3 に複写します.

# C2: =VLOOKUP(B2,\$A\$10:\$B\$18,2,FALSE) 複写元: C2 複写先: C3

## 11.4 2駅間距離を求める(絶対値,切り上げ)

2 つの駅間の距離を求めるには, 2 つの駅の川崎からの距離の差を求めます. 負の値になることがあるので,絶対値をとり ます. 絶対値をとる関数は, ABS です. まず, 2 駅間距離切り上げ前は,

C4: =ABS( C2 - C3 )

となります.JRの運賃計算は、小数点以下の端数は切り上げです.そこで、C4の値を切り上げて、C5に表示します.

C5: **=**ROUNDUP(C4,0)

となります. ROUNDUP(A,B) で, A は切り上げる数, B は切り上げる単位を示しています. 0 で小数点の位置, 1 で 0.1 単位, 2 で 0.01 単位, -1 で 10 の単位で切り上げです. 同様に, ROUND が四捨五入, ROUNDDOWN が切り捨てです (表 3).

表3 切り上げ,四捨五入,切り捨て

$\text{ROUNDUP}(3.14,0) \rightarrow 4$	$ROUNDUP(586,-1) \rightarrow 590$	$\text{ROUNDUP}(16.58,1) \rightarrow 16.6$
$\text{ROUND(3.14,0)} \rightarrow 3$	$\text{ROUND(586,-1)} \rightarrow 590$	$\text{ROUND}(16.58,1) \rightarrow 16.6$
ROUNDDOWN(3.14,0) $\rightarrow$ 3	ROUNDDOWN(586,-1) $\rightarrow$ 580	ROUNDDOWN(16.58,1) $\rightarrow$ 16.5

## 12 練習問題

## 12.1 練習問題 S102-1

図 20 は、小田急ロマンスカーの新宿からの運賃,特急料金の表です(路線図). B14 に駅名を入力したら、B15 の運賃, B16 の特急料金を計算する式を設定しましょう. この表は、テキストのページからダウンロードできます(S102.xlsx(ロマン スカー)).

	А	В	С	
1	新宿からの	ロマンスカー	ー運賃·特急	急料金
2		運賃(切符)	特急料金	
3	向ヶ丘遊園	250	300	
4	新百合ヶ丘	310	410	
5	町田	370	410	
6	相模大野	370	410	
7	本厚木	500	570	
8	秦野	670	620	
9	新松田	780	690	
10	小田原	880	890	
11	箱根湯本	1190	890	
12				
13	新宿からロ	マンスカーる	を利用(切符	F)
14	降車駅	秦野		
15	運賃			
16	特急料金			
17	合計			

図 20 ロマンスカーの運賃・料金

ヒント: B16 を計算するとき, A3:C11 までを照合範囲として, Vlookup 関数を使います. 列位置を設定することで特急 料金を表示させます.

#### 12.2 練習問題 S4-4

S4.xlsx のモーグルで,「エア点」の計算と「スピード」の計算では,小数第3桁以下を切り捨て,小数点以下第2位まで 求めるように変更ししましょう.

## 12.3 練習問題 S103-1

図 22 は,表 21 にしたがって.得点から,S,A+,A,B+,B,C+,C,Fの評価を計算する表です.VLOOKUPで使う得点と評価の対応表(照合範囲)を作成し,C2~C11の計算式を設定しなさい.この表は,テキストのページからダウンロードできます(S103.xlsx(成績評価)).

得点	評価
90~100	S
85~89	A+
80~84	А
75~79	B+
70~74	В
65~69	C+
60~64	С
0~59	F

	Α	В	С	D
1	名前	得点	評価	
2	田中	100	S	
3	大西	70	В	
4	大久保	30	D	
5	前田	80	А	
6	友田	65	C+	
7	保田	77	B+	
8	大川	78	B+	
9	木村	32	D	
10	村田	54	D	
11	川上	66	C+	
12				

#### 図 22 得点から成績を求める

図 21 大学の成績評価

VLOOKUP 関数で照合方法が TRUE(行と行との間も検索する場合)のとき、照合値の列(一番左の列)が昇順(小さい順)でソート(並べ替え)されていなくてはなりません.したがって、表 21 に対応する VLOOKUP の照合範囲の表は、表 21 の行の順番とは逆になります.「0 から 59」の行が1 行目、「90~」の行が最下行になります.

## 13 絶対参照と相対参照の混合とそのスピルでの計算

図 23 は, ネットショッピングの支払いを各通貨別に集計したものです. 換算レートは, 2012 年 3 月現在のものです. 赤の部分は計算式で求める部分です. この表は, テキストのページからダウンロードできます (S104.xlsx(ネットショッピング)).

表には、日付と品名、支払った通貨の種類(GBP はイギリスポンド、EUR はユーロ、CHF はスイスフラン、USD はアメ リカドルをあらわします.)、その通貨での金額を記入します. 表の右側で、通貨毎に集計し、換算レートをかけて、日本円 (邦貨)の金額を求めます. 日付の入力では「8 月 10 日」は「8/10」と入力します. 自動的に、「8 月 10 日」に変換されます. 日付としてうまく表示されない場合や、日付から数値の表示形式に変更するときは、右クリックして、セルの書式設定 の表示形式 のタブで変更します.

最終的には、図 23 のような表にして、各通貨毎の合計等を求めます.

まず,計算式を設定しましょう.まず,E4 について考えます.

E4 は、もし、この行の通貨が、GBP であったら、この行の金額の欄の値を表示する.

Ļ

E4 は,もし,C4 と E3 が等しかったら,D4 を表示し,そうでなかったら,空白を表示する (空白は、ダブルコーテーション「"」を2つ重ねます).

	А	В	С	D	Е	F	G	Н	Ι
1	ネットショッ	ピング							
2									
3	日付	品名	通貨	金額	GBP	EUR	CHF	USD	
4	8月10日	洋服	GBP	85	85				
5	8月10日	バッグ	GBP	70	70				
6	8月11日	バッグ	EUR	1000		1000			
7	8月13日	洋服	EUR	500		500			
8	8月13日	バッグ	EUR	600		600			
9	8月14日	書籍	EUR	150		150			
10	8月14日	洋服	CHF	150			150		
11	8月14日	書籍	USD	100				100	
12	8月15日	ソフトウエア	USD	130				130	
13		合計			155	2250	150	230	
14		換算レート			132.15	109.88	91.08	83.41	合計
15		邦貨金額(日本	円の金額)		20,483	247,230	13,662	19,184	300,560

図 23 ネットショッピング支払い

## E4: =IF(C4=E3,D4,"")

となります.次に、この計算式を複写してみましょう.

としてみましょう. 間違っていることがわかります.

このような場合,2つの方法で解決できます.1つはスピルを使った方法で,もう1つは絶対参照と相対参照の混合させる方法です.

#### 13.1 スピルを使った計算

スピルを使っての計算式の場合, E4 の =IF(C4=E3,D4,"") の IF 関数の各部分について範囲 E4:H12 の全セルに対して, 図 24 のように考えます. 範囲 D(出力の範囲) は, 範囲 A と範囲 B の値が等しいとき, 範囲 C の値を出力します. そうでは ないとき, ""を出力します.



図 24 スピルによる計算, ネットショッピング支払い

IF の条件部 この部分は、C列(C4:C12範囲 A)と 3行目(E3:H3範囲 B)の値が等しかったらという条件になるので、

C4:C12=E3:H3 とします.

条件を満たしているとき この部分は, D 列 (D4:D12 範囲 C) の値を表示するので, D4:D12 とします. 条件を満たしていいないとき ("") そのまま "" とします

## E4: =IF(C4:C12=E3:H3,D4:D12,"")

とします.

動画:スピル・ネットショッピング(音声付き)・動画:スピル・ネットショッピング(音声なし)

## 13.2 絶対参照の利用(絶対参照と相対参照の混合を使った計算)

61

E5,E6 の計算式(表 4 上部)を見てみましょう. E5 の計算式は, IF(C5=E4,D5,"")になっています. しかし, 正しい計算 式は, IF(C5=E3,D5,"")です. E6,E7 も表のようになります. 原因は, E4 の式中の E3 は, 下に複写するとき, 数字の部分 が増えてはいけないのに, 番地調整機能によって, 増えてしまっているからです.

セル	単純に 後 与 し に と さ の 計 昇 式	止しい計昇式
E4:	=IF(C4=E <mark>3</mark> ,D4,"")	=IF(C4=E3,D4,"")
E5:	=IF(C5=E4,D5,"")	=IF(C5=E3,D5,"")
E6:	=IF(C6=E5,D6,"")	=IF(C6=E3,D6,"")
E7:	=IF(C7=E <mark>6</mark> ,D7,"")	=IF(C7=E3,D7,"")
E4:	=IF(C4=E3,D4,"")	=IF(C4=E3,D4,"")
F4:	=IF( <b>D</b> 4=F3, <b>E</b> 4,''')	=IF(C4=F3,D4,"")
G4:	=IF( <b>E</b> 4=G3, <b>F</b> 4,'''')	=IF(C4=G3,D4,"")
H4:	=IF( <b>F</b> 4=H3, <b>G</b> 4,''')	=IF(C4=H3,D4,"")

表4 絶対参照と相対参照の混合

出始に指定したしたの計算子 エレン計算子

E4 の計算式を右に複写したものを表 4 下部に示します.赤の部分が間違っています.これは, C4 の部分と D4 の部分の

アルファベットは,横に複写されるときに,増えてはいけないのに番地調整機能により,増えてしまったからです. 複写のとき,番地調整機能を働かせないように「\$」を付けます.上の例では,

## E4: =IF(\$C4=E\$3,\$D4,"")

となります. 複写しても, C と 3 と D は, 変化せずいつも C と 3 と D です. E4 を設定したら, 他の部分に複写します.

複写元: E4 複写先: E4:H12

#### 13.3 日本円への換算等

合計などを計算しましょう.換算レートは、1 海外通貨あたり日本円でいくらかを示しています.例えば、155 ポンド (GBP) 使い、換算レートが1 ポンド=132.154 円のときの邦貨金額 (日本円金額) は、155 ポンド × 132.15 円/ポンド = 20,483 円 となります. I15 は、邦貨金額の合計です. 15 行目はセルの書式設定で、小数点以下を四捨五入しましょう. 通貨記号の欄を変更したり、現在の為替レート (朝日新聞デジタル:為替 - 経済・マネー) に直したりしても正しく動作

するか確認しましょう.

## 14 練習問題

### 14.1 1次関数3本のグラフ化

1 次関数 f(x) = ax + bで,  $a \ge b$ を変えた 3 本をグラフ化する表を作成しましょう. 図 25 左の B2:D3 に 3 本の 1 次関数の  $a \ge b$ の値を記入しておきます. ただし,計算式を設定する部分は,次のようにします.

(1) A6 に -10 を入力し、A7 に「=A6+0.5」とします.

(2) A7の計算式を A46 まで複写します.

(3) B6:D46の計算式:絶対参照を利用しもスピルを利用してよい.

- B6 に絶対参照と相対参照が混在した計算式を設定し、
- その式を B6:D46 に複写します.
- (4) スピル
  - B6:D46 を範囲指定して、まとめて計算式を設定します.
  - x の値の範囲: A6:A46
  - a の値の範囲: B2:D2
  - b の値の範囲: 考えてみてください



図 25 y = ax + bのグラフ化のための表とグラフ

(5) 図 25 右のような 3 本のグラフ(散布図(直線))を作成しましょう.

(6) それぞれの直線のパラメータである a,b を変化させたとき,どのように変化するのかを確かめましょう.

	А	В	C	D	E	F	G	Н	Ι
1									
2	日	通話区分	距離	通話時間 (秒)	10円で話せ る時間	度数 (切り上げ前)	度数 (切り上げ後)	金額	
3	6月10日	1	35	588	45	13.06666667	14	140	
4	6月11日	2	48	627	60	10. 45	11	110	
5	6月12日	3	165	37	45	0.822222222	1	10	
6	6月12日	3	165	156	45	3. 466666667	4	40	
7	6月13日	2	23	251	60	4. 183333333	5	50	
8	6月15日	1	18	1028	90	11. 42222222	12	120	
9							合計	470	
10									
11									
12	料金表(	<u>1 0円で話せる</u>	5時間)						
13	距離	1 平日昼間	2 夜・休日昼夜	3 深夜早朝					
14	0	90	90	120					
15	20	60	60	75					
16	30	45	60	75					
17	60	26	45	60					
18	100	20	26	45					
19	170	20	26	36					

図26 電話料金の計算

### 14.2 練習問題 S105-1

図 26 は、ある架空の電話会社の電話料金の計算表です.この表は、テキストのページからダウンロードできます (S105.xlsx(電話料金)).

回答は、絶対参照と相対参照を混在させた式で設定してください、スピルでは、スピル範囲の値に計算式を追加できないよ

- •10円で話せる時間は、通話区分、距離で決まります。
- 通話区分は3区分:「平日の昼間」(区分1),「平日の夜または休日の昼夜」(区分2),「平日,休日の深夜」(区分3)
- 距離は,通話先までの距離で,表の14行目の0は,0km(隣接)以上20km未満を表しています.また,19行目の距離 170の行は,170km以上すべてを表しています.
- たとえば,通話区分が2で,70kmの場合,60の行の通話区分2の列(3列目)の場合の10円で話せる秒数は45秒 になります.
- 料金は、まず度数を、

## 度数 = <u>通話時間(秒)</u> 10 円で話せる秒数

で計算し (F列),小数点以下を切り上げます (G列,切り上げの計算方法は,表3を参照).

•1 度数は 10 円なので, 度数の列 (G 列) に 10 をかけたものが電話料金になります.

では、図 26 の上方のエリアを使って電話料金の計算表を作成しましょう. ヒント

- •「10 円で話せる秒数」は、距離と通話区分を使って、料金表 (A14:D19) から VLOOKUP を使って探します. そのとき,通話区分が1のとき2列目,2のとき3列目...であることから、VLOOKUPの列位置は、通話区分のセルの値を使った計算式になります (通話区分の値に1加えた数が列番号になっています).
- •「度数 (切り上げ前)」は、単純に、<u>通話時間(秒)</u>を計算したものです.
- •「度数(切り上げ後)」は、「度数(切り上げ前)」の値を切り上げたものです.

完成したら,通話区域,通話の種類や時間を変えてみて,正しく計算されているかどうかチェックしましょう.